
按单元格读取WORD表格数据
编辑：madmlwt

1．按单元格读取表格数据到当前工作表
【问题】

与被操作的工作簿保存在相同位置的文件夹“WORD表格”中的文档“工程订单.doc”，如下图所示：

把表格中的数据从“序号”单元格开始逐个单元格读取到被操作的EXCEL工作簿的当前工作表A1开始的单元格区域。

【代码1】

Sub逐个单元格读取WORD表格数据() 

Application.ScreenUpdating = False '关闭屏幕刷新

开始时间 = Timer '秒

wjM = ThisWorkbook.Path & "\WORD表格\工程订单.doc"

ActiveSheet.Cells.ClearContents '清除表中已有的数据内容

Set wDoc = CreateObject(wjM) '定义Word对象

Set wTable = wDoc.Tables(1) '定义表格对象

hs = wTable.Rows.Count '计算表格总行数

ls = wTable.Columns.Count '计算表格总列数

For i = 1 To hs '从表格第1行循环到最后一行

    For j = 1 To ls '从表格第1列循环到最后一列

        x = wTable.Cell(i, j).Range.Text '单元格对象(包含回车符)

        ActiveSheet.Cells(i, j) = Left(x, Len(x) - 1) '把去掉回车符后的文本写入EXCEL单元格

    Next j

Next i

wDoc.Close '关闭文档

Set wTable = Nothing '释放表格对象变量存储空间

Set wDoc = Nothing '释放文档对象变量存储空间

ActiveSheet.Cells.EntireColumn.AutoFit '自动调整列宽

MsgBox Chr(10) & "成功读取文件 " & Dir(wjM) & " 中表格数据到当前工作表" _

       & Chr(10) & Chr(10) & "读取数据共用 " & Timer - 开始时间 & " 秒", , "逐个单元格读取WORD表格数据"

Application.ScreenUpdating = False '关闭屏幕刷新

End Sub
说明：
写入EXCEL中数据保持原表格中数据格式，所有数据都是文本格式，不能进行计算。

【代码2】

Sub逐个单元格读取WORD表格数据() 

Application.ScreenUpdating = False '关闭屏幕刷新

开始时间 = Timer '秒

wjM = "工程订单"

docM = ThisWorkbook.Path & "\WORD表格\" & wjM & ".doc" '被引用的DOC文件

If Dir(docM) = wjM & ".doc" Then '如果指定引用的文件存在

   ActiveSheet.Cells.ClearContents '清除当前工作表中所有数据内容

   Set wdApp = CreateObject(docM) '定义一个Word对象变量

   With wdApp.Tables(1) '打开的WORD文件第一个表格

        hs = .Rows.Count '表格的行数

        ls = .Columns.Count '表格的列数

        For h = 1 To hs '从第1行循环到最后一行

            For l = 1 To ls '从第1列循环到最后一列

                If l = 1 Then'第1列设置为文本格式
                   ActiveSheet.Cells(h, l).NumberFormatLocal = "@"

                   ActiveSheet.Cells(h, l) = Replace(.Cell(h, l).Range.Text, Chr(13) & Chr(7), "") 'WORD表格中单元格字符去掉回车符和黑加点的数据写入工作表单元格(Che$(13)=Chr(13))

                ElseIf l > 1 Then'第1列后的列只读取数据值
                   ActiveSheet.Cells(h, l) = Replace(.Cell(h, l).Range.Text, Chr(13) & Chr(7), "") 'WORD表格中单元格字符去掉回车符和黑加点的数据写入工作表单元格

                End If

            Next l

        Next h

   End With

   wdApp.Close '关闭文件

   Set wdApp = Nothing '释放存储空间

   ActiveSheet.Cells.EntireColumn.AutoFit '自动调整列宽

   MsgBox Chr(10) & "成功地把 " & Dir(docM) & " 表格中数据导入当前工作表 ！" _

        & Chr(10) & Chr(10) & "共用时 " & Timer - 开始时间 & " 秒", , "按单元格读取WORD表格数据"

ElseIf Dir(docM) = "" Then '如果指定引用的文件不存在

   MsgBox Chr(10) & "指定引用的文件 " & Dir(wjM) & " 不存在！", , "按单元格读取WORD表格数据"

   Exit Sub '退出程序

End If

Application.ScreenUpdating = False '关闭屏幕刷新

End Sub
说明：

单独设置如0打头或身份证号码类数据列在EXCEL中的单元格格式，导入EXCEL的数据可以进行计算。

2．读取多表格数据到当前工作表

【问题】

与被操作的工作簿保存位置相同的文件夹“WORD表格”中的文档“订单.doc”中有多个格式完全相同的表格，把这些表格中数据逐个单元格读取到被操作的当前工作表A1起始的单元格区域。
【代码】

Sub读取多表格数据到当前工作表()

Application.ScreenUpdating = False '关闭屏幕刷新

dqT = Timer '程序运行开始时间(秒)

wjM = "订单"

docM = ThisWorkbook.Path & "\WORD表格\" & wjM & ".doc" '被引用的DOC文件

If Dir(docM) = wjM & ".doc" Then '如果指定引用的文件存在

   ActiveSheet.Cells.ClearContents '清除当前工作表中所有数据内容

   Set wdApp = CreateObject(docM) '定义一个Word对象变量

   gs = wdApp.Tables.Count '文档中表格总个数

   For g = 1 To gs

       With wdApp.Tables(g) '打开的WORD文件第一个表格

            hs = .Rows.Count '表格的行数

            If g = 1 Then

               起始行 = 1

            ElseIf g > 1 Then

               起始行 = 2

            End If

            For h = 起始行 To hs '从起始行循环到最后一行

                l = 1 '起始单元格个数

                If g = 1 Then

                   xrh = h '数据写入工作表的行序号

                ElseIf g > 1 Then

                   xrh = ActiveSheet.Cells(Rows.Count, 1).End(xlUp).Row + 1 '数据写入工作表的行序号

                End If

                For Each sjH In .Rows(h).Cells '遍历WORD表格的每行每个单元格

                    If l = 1 Then

                       ActiveSheet.Cells(xrh, l).NumberFormatLocal = "@" '文本格式

                       ActiveSheet.Cells(xrh, l) = Replace(.Cell(h, l).Range.Text, Chr(13) & Chr(7), "") 'WORD表格中单元格字符去掉回车符和黑加点的数据写入工作表单元格

                    ElseIf l > 1 Then

                       ActiveSheet.Cells(xrh, l) = Replace(.Cell(h, l).Range.Text, Chr(13) & Chr(7), "") 'WORD表格中单元格字符去掉回车符和黑加点的数据写入工作表单元格(Che$(13)=Chr(13))

                    End If

                    l = l + 1 '单元格个数计数器

                Next sjH

            Next h

       End With

   Next g

   wdApp.Close '关闭文件

   Set wdApp = Nothing '释放存储空间

   ActiveSheet.Cells.EntireColumn.AutoFit '自动调整列宽

   MsgBox Chr(10) & "成功地把 " & Dir(docM) & " 中的 " & gs & " 个表格中数据导入当前工作表 ！" _

        & Chr(10) & Chr(10) & "共用时 " & Timer - dqT & " 秒", , "读取WORD写入EXCEL"

ElseIf Dir(docM) = "" Then '如果指定引用的文件不存在

   MsgBox Chr(10) & "指定引用的文件 " & wjM & ".doc 不存在！", , "读取WORD写入EXCEL"

   Exit Sub '退出程序

End If

Application.ScreenUpdating = False '关闭屏幕刷新

End Sub
说明：

WORD文档中有多个格式完全相同的表格，则可以先计算出表格个数，再按表格个数进行循环读取。或者采取遍历文档中的表格循环方式处理亦可。
可以根据需要灵活设置某列的数据格式，比如0打头的数字或身份证号码等，一定要设置EXCEL中对应列的单元格格式为“文本”。

Chr(13) ——回车符，Chr(7)——小黑点。

